Pensieve header: A first implementation of nilpotent integration.

Initialization

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank"];
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/icbs24 to compute rotation numbers.
```

In[∘]:=
```
CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ CCF[c] (Times @@ vs^ps)] ];
```

Integration

In[∘]:=
```
Unprotect[Integrate];
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, Q, G, V, s, t, k, a, b},
   n = Length@vs;
   Q = -Table[(∂_{vs⟦a⟧,vs⟦b⟧} L) /. Thread[vs → 0], {a, n}, {b, n}];
   G = Inverse[Q] / 2;
   V = L + vs.Q.vs / 2;
   s = t = V; k = 0;
   While[0 =!= t,
    s +=
     1/(++k)! (t = CF@Sum[G⟦a, b⟧ ((∂_{vs⟦a⟧,vs⟦b⟧} t) + (∂_{vs⟦a⟧} t) (∂_{vs⟦b⟧} t)), {a, n}, {b, n}])];
   PowerExpand@Factor[ω (Det[Q] (2 π)^n)^(-1/2)] × 𝔼[CF@s /. Thread[vs → 0]]
  ];
Protect[Integrate];
```

In[∘]:=
$$\int \mathbb{E}\left[i \lambda x_1^2 / 2\right] d\{x_1\}$$

Out[∘]=
$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\ \sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-i\,\lambda\,x_1^2/2\right] \mathrm{d}\{x_1\}$

Out[∘]=

$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] \mathrm{d}\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda\,x_1^2/2\right] \mathrm{d}\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-x_1^2/2 + \xi\,x_1\right] \mathrm{d}\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] \mathrm{d}\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\left(-b^2 + a\,c\right)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] \mathrm{d}\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{\left(-b^2 + a\,c\right)x_2^2}{2\,a} + \frac{\xi_1^2}{2\,a} + \frac{x_2\left(-b\,\xi_1 + a\,\xi_2\right)}{a}\right]}{\sqrt{a}\,\sqrt{2\pi}}$$

In[∘]:= $\int I1\, \mathrm{d}\{x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\left(-b^2 + a\,c\right)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{y_1, y_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{y_1, y_2\} + \{\eta_1, \eta_2\}.\{y_1, y_2\}\right] d\{y_1, y_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\eta_1^2 - 2\,b\,\eta_1\,\eta_2 + a\,\eta_2^2}{2\,(-b^2 + a\,c)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{y_1, y_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{y_1, y_2\} + \{\eta_1, \eta_2\}.\{y_1, y_2\}\right] d\{y_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{b^2\,y_2^2 - a\,c\,y_2^2 - 2\,b\,y_2\,\eta_1 + \eta_1^2 + 2\,a\,y_2\,\eta_2}{2\,a}\right]}{\sqrt{a}\;\sqrt{2\,\pi}}$$

In[∘]:= $\int I1\, d\{y_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{a\,c\,\eta_1^2 - 2\,a\,b\,\eta_1\,\eta_2 + a^2\,\eta_2^2}{2\,a\,(-b^2 + a\,c)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[\xi\,x + \eta\,y + z\,(x - y) + x^2\right] d\{x, z\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}[y\,(y + \eta + \xi)]}{2\,\pi}$$

The ρ_1 Integrand

Adopted from pensieve://Projects/APAI/PerturbedGaussianIntegration.nb.

In[∘]:=

```
q[s_, i_, j_] := x_i (p_i - T^s p_{i+1} + (T^s - 1) p_{j+1}) + x_j (p_j - p_{j+1});
r_1[s_, i_, j_] :=
  s (-1 + 2 p_i x_i - 2 p_j x_i + (-1 + T^s) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2;
ℒ[s_, i_, j_] := -q[s, i, j] + ϵ r_1[s, i, j];
γ_1[φ_, k_] := ϵ φ (1 / 2 - x_k p_k);
ρ1i[K_] := Module[{Cs, φ, n, s, i, j, k, vs, L},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   L = -x_{2 n+1} p_{2 n+1};
   Cases[Cs, {s_, i_, j_} :> (L += ℒ[s, i, j])];
   L += ϵ Sum[γ_1[φ[[k]], k], {k, 2 n}];
   CF@(L + O[ϵ]^2)
  ];
ρ1vs[K_] := Union @@ Table[{x_i, p_i}, {i, 2 Crossings[K] + 1}]
```

In[∘]:=
```
ρ1i[Knot[3, 1]]
```

Out[∘]=

$$\left(-p_1 x_1 + p_2 x_1 - p_2 x_2 + \frac{p_3 x_2}{T} + \frac{(-1+T)\, p_6 x_2}{T} - p_3 x_3 + p_4 x_3 + \frac{(-1+T)\, p_2 x_4}{T} - p_4 x_4 + \frac{p_5 x_4}{T} - p_5 x_5 + p_6 x_5 + \frac{(-1+T)\, p_4 x_6}{T} - p_6 x_6 + \frac{p_7 x_6}{T} - p_7 x_7\right) + \left(1 - p_2 x_2 + p_5 x_2 + \frac{(-1+T)\, p_2 p_5 x_2^2}{2T} - \frac{(-1+T)\, p_5^2 x_2^2}{2T} + p_1 x_4 - p_1^2 x_1 x_4 + p_1 p_4 x_1 x_4 - \frac{(-1+T)\, p_1^2 x_4^2}{2T} + \frac{(-1+T)\, p_1 p_4 x_4^2}{2T} + p_2 p_5 x_2 x_5 - p_5^2 x_2 x_5 + p_3 x_6 - p_6 x_6 - p_3^2 x_3 x_6 + p_3 p_6 x_3 x_6 - \frac{(-1+T)\, p_3^2 x_6^2}{2T} + \frac{(-1+T)\, p_3 p_6 x_6^2}{2T}\right) \epsilon + O[\epsilon]^2$$

In[∘]:=
```
ρ1vs[Knot[3, 1]]
```

Out[∘]=

$$\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, x_1, x_2, x_3, x_4, x_5, x_6, x_7\}$$

Integration of ϵ-Series

In[∘]:=
```
Unprotect[Integrate];
∫ω_. 𝔼[L_SeriesData] ⅆ(vs_List) := Module[{n, L0, Q, Δ, G, V, s, t, k, a, b},
   n = Length@vs; L0 = Normal@L /. ϵ → 0;
   Q = -Table[(∂_{vs[[a]],vs[[b]]} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = CF@Det[Q]) == 0,
    Return["How dare you ask me to integrate a singular Gaussian!"]];
   G = Inverse[Q] / 2;
   V = L + vs.Q.vs / 2;
   s = t = V; k = 0;
   While[0 =!= Normal@t,
    s +=
     1/(++k)! (t = CF@Sum[G[[a, b]] ((∂_{vs[[a]],vs[[b]]} t) + (∂_{vs[[a]]} t) (∂_{vs[[b]]} t)), {a, n}, {b, n}])];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] × 𝔼[CF@s /. Thread[vs → 0]]
  ];
Protect[Integrate];
```

In[∘]:=
```
∫𝔼[x₁ p₁ + ϵ x₁^7 p₁^7 + O[ϵ]^2] ⅆ{x₁, p₁}
```

Out[∘]=

$$-\frac{i\, \mathbb{E}\left[-5040\, \epsilon + O[\epsilon]^2\right]}{2\pi}$$

In[∘]:=
$\int \mathbb{E}\left[x_1 p_2 + \epsilon\, x_2^7 p_1^7 + O[\epsilon]^2\right] ⅆ\{x_1, p_2\}$

Out[∘]=
$$-\frac{i\, \mathbb{E}\left[p_1^7 x_2^7 \epsilon + O[\epsilon]^2\right]}{2\pi}$$

In[∘]:=
$\int \mathbb{E}\left[x_1 p_2 + 3\, x_2 p_1 + \epsilon\, p_2^5 x_1^5 + O[\epsilon]^2\right] ⅆ\{x_1, x_2, p_1, p_2\}$

Out[∘]=
$$\frac{\mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{12\pi^2}$$

In[∘]:=
$\int \mathbb{E}\left[x_1 p_2 + x_2 p_3 + x_3 p_1 + \epsilon\, x_1^5 p_2^5 + O[\epsilon]^2\right] ⅆ\{x_1, x_2, x_3, p_1, p_2, p_3\}$

Out[∘]=
$$-\frac{i\, \mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{8\pi^3}$$

In[∘]:= MatrixForm@Table[
$\int \mathbb{E}[x_1 p_2 + x_2 p_3 + x_3 p_1 + \xi_i x_i + \pi_j p_j]\, ⅆ\{x_1, x_2, x_3, p_1, p_2, p_3\}$,
{i, 3}, {j, 3}]

Out[∘]//MatrixForm=
$$\begin{pmatrix} -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2 \xi_1]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3 \xi_2]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1 \xi_3]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \end{pmatrix}$$

```
In[•]:= K = Knot[5, 2];
        {ρ1i@K, ρ1vs@K}
        ∫𝔼[ρ1i@K] ⅆ(ρ1vs@K)
```

Out[•]=

$$\left\{\left(-p_1 x_1+p_2 x_1-p_2 x_2+\frac{p_3 x_2}{T}+\frac{(-1+T)\, p_8 x_2}{T}-p_3 x_3+p_4 x_3+\frac{(-1+T)\, p_2 x_4}{T}-p_4 x_4+\frac{p_5 x_4}{T}-p_5 x_5+p_6 x_5-p_6 x_6+\frac{p_7 x_6}{T}+\frac{(-1+T)\, p_{10} x_6}{T}-p_7 x_7+p_8 x_7+\frac{(-1+T)\, p_4 x_8}{T}-p_8 x_8+\frac{p_9 x_8}{T}-p_9 x_9+p_{10} x_9+\frac{(-1+T)\, p_6 x_{10}}{T}-p_{10} x_{10}+\frac{p_{11} x_{10}}{T}-p_{11} x_{11}\right)+\left(2-p_2 x_2+p_7 x_2+\frac{(-1+T)\, p_2 p_7 x_2^2}{2T}-\frac{(-1+T)\, p_7^2 x_2^2}{2T}+p_1 x_4-p_1^2 x_1 x_4+p_1 p_4 x_1 x_4-\frac{(-1+T)\, p_1^2 x_4^2}{2T}+\frac{(-1+T)\, p_1 p_4 x_4^2}{2T}-p_6 x_6+p_9 x_6+\frac{(-1+T)\, p_6 p_9 x_6^2}{2T}-\frac{(-1+T)\, p_9^2 x_6^2}{2T}+p_2 p_7 x_2 x_7-p_7^2 x_2 x_7+p_3 x_8-p_8 x_8-p_3^2 x_3 x_8+p_3 p_8 x_3 x_8-\frac{(-1+T)\, p_3^2 x_8^2}{2T}+\frac{(-1+T)\, p_3 p_8 x_8^2}{2T}-p_9 x_9+p_6 p_9 x_6 x_9-p_9^2 x_6 x_9+p_5 x_{10}-p_5^2 x_5 x_{10}+p_5 p_{10} x_5 x_{10}-\frac{(-1+T)\, p_5^2 x_{10}^2}{2T}+\frac{(-1+T)\, p_5 p_{10} x_{10}^2}{2T}\right)\epsilon+O[\epsilon]^2,\right.$$
$$\left.\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, p_8, p_9, p_{10}, p_{11}, x_1, x_2, x_3, x_4, x_5, x_6, x_7, x_8, x_9, x_{10}, x_{11}\}\right\}$$

Out[•]=

$$-\frac{i\, T^4\, \mathbb{E}\left[\frac{(-1+T)^2\,(5-4T+5T^2)\,\epsilon}{(2-3T+2T^2)^2}+O[\epsilon]^2\right]}{2048\,\pi^{11}\,(2-3T+2T^2)}$$

In[∘]:=
```
K = Knot[8, 19];
{ρ1i@K, ρ1vs@K}
∫𝔼[ρ1i@K] ⅆ(ρ1vs@K)
```

Out[∘]=

$$\{(-p_1 x_1 + T p_2 x_1 + (1-T) p_5 x_1 - p_2 x_2 + p_3 x_2 - p_3 x_3 + T p_4 x_3 + (1-T) p_9 x_3 - p_4 x_4 + p_5 x_4 - p_5 x_5 + p_6 x_5 - p_6 x_6 + T p_7 x_6 + (1-T) p_{14} x_6 + (1-T) p_3 x_7 - p_7 x_7 + T p_8 x_7 - p_8 x_8 + p_9 x_8 - p_9 x_9 + p_{10} x_9 - p_{10} x_{10} + T p_{11} x_{10} + (1-T) p_{16} x_{10} - p_{11} x_{11} + p_{12} x_{11} + (1-T) p_6 x_{12} - p_{12} x_{12} + T p_{13} x_{12} - p_{13} x_{13} + p_{14} x_{13} + (1-T) p_{10} x_{14} - p_{14} x_{14} + T p_{15} x_{14} - p_{15} x_{15} + p_{16} x_{15} + (1-T) p_{12} x_{16} - p_{16} x_{16} + T p_{17} x_{16} - p_{17} x_{17}) + \Big(-4 + p_1 x_1 - p_4 x_1 + \frac{1}{2}(-1+T) p_1 p_4 x_1^2 + \frac{1}{2}(1-T) p_4^2 x_1^2 + p_3 x_3 - p_8 x_3 + \frac{1}{2}(-1+T) p_3 p_8 x_3^2 + \frac{1}{2}(1-T) p_8^2 x_3^2 + p_4 x_4 - p_1 p_4 x_1 x_4 + p_4^2 x_1 x_4 + p_6 x_6 - p_{13} x_6 + \frac{1}{2}(-1+T) p_6 p_{13} x_6^2 + \frac{1}{2}(1-T) p_{13}^2 x_6^2 - p_2 x_7 + p_7 x_7 + p_2^2 x_2 x_7 - p_2 p_7 x_2 x_7 + \frac{1}{2}(1-T) p_2^2 x_7^2 + \frac{1}{2}(-1+T) p_2 p_7 x_7^2 - p_3 p_8 x_3 x_8 + p_8^2 x_3 x_8 + p_{10} x_{10} - p_{15} x_{10} + \frac{1}{2}(-1+T) p_{10} p_{15} x_{10}^2 + \frac{1}{2}(1-T) p_{15}^2 x_{10}^2 - p_5 x_{12} + p_5^2 x_5 x_{12} - p_5 p_{12} x_5 x_{12} + \frac{1}{2}(1-T) p_5^2 x_{12}^2 + \frac{1}{2}(-1+T) p_5 p_{12} x_{12}^2 - p_6 p_{13} x_6 x_{13} + p_{13}^2 x_6 x_{13} - p_9 x_{14} + p_{14} x_{14} + p_9^2 x_9 x_{14} - p_9 p_{14} x_9 x_{14} + \frac{1}{2}(1-T) p_9^2 x_{14}^2 + \frac{1}{2}(-1+T) p_9 p_{14} x_{14}^2 - p_{10} p_{15} x_{10} x_{15} + p_{15}^2 x_{10} x_{15} - p_{11} x_{16} + p_{16} x_{16} + p_{11}^2 x_{11} x_{16} - p_{11} p_{16} x_{11} x_{16} + \frac{1}{2}(1-T) p_{11}^2 x_{16}^2 + \frac{1}{2}(-1+T) p_{11} p_{16} x_{16}^2\Big) \epsilon + O[\epsilon]^2,$$
$$\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, p_8, p_9, p_{10}, p_{11}, p_{12}, p_{13}, p_{14}, p_{15}, p_{16}, p_{17}, x_1, x_2, x_3, x_4, x_5, x_6, x_7, x_8, x_9, x_{10}, x_{11}, x_{12}, x_{13}, x_{14}, x_{15}, x_{16}, x_{17}\}\}$$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-\frac{(-1+T)^2 (1+T^4)(3+4T^3+3T^6)\epsilon}{(1-T+T^2)^2 (1-T^2+T^4)^2} + O[\epsilon]^2\right]}{131072\,\pi^{17}\,T\,(1-T+T^3-T^5+T^6)}$$

Invariance Under Reidemeister 2b

In[∘]:=
```
lhs =
  ∫𝔼[π_i p_i + π_j p_j + ℒ[1, i, j] + ℒ[-1, i + 1, j + 1] + O[ϵ]^2] ⅆ{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}}
```

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\,\pi_i + p_{2+j}\,\pi_j) + O[\epsilon]^2\right]}{16\,\pi^4}$$

In[∘]:= $\text{rhs} = \int \mathbb{E}[\pi_i p_i + \pi_j p_j + x_i (p_{i+1} - p_i) + x_{i+1} (p_{i+2} - p_{i+1}) + x_j (p_{j+1} - p_j) + x_{j+1} (p_{j+2} - p_{j+1}) + O[\epsilon]^2]$
$d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=
$$\frac{\mathbb{E}[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\epsilon]^2]}{16\pi^4}$$

In[∘]:= lhs == rhs

Out[∘]=
True

Invariance Under R2c

In[∘]:= $\text{lhs} = \int \mathbb{E}[\pi_i p_i + \pi_j p_j + \mathcal{L}[-1, i, j+1] + \mathcal{L}[1, i+1, j] + \gamma_1[-1, j+1] + O[\epsilon]^2]$
$d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=
$$\frac{\mathbb{E}[(p_{2+i}\pi_i + p_{2+j}\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2]}{16\pi^4}$$

In[∘]:= $\text{rhs} = \int \mathbb{E}[\pi_i p_i + \pi_j p_j + x_i (p_{i+1} - p_i) + x_{i+1} (p_{i+2} - p_{i+1}) + x_j (p_{j+1} - p_j) +$
$x_{j+1} (p_{j+2} - p_{j+1}) + \gamma_1[-1, j+1] + O[\epsilon]^2]\, d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=
$$\frac{\mathbb{E}[(p_{2+i}\pi_i + p_{2+j}\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2]}{16\pi^4}$$

In[∘]:= lhs == rhs

Out[∘]=
True

Invariance Under Reidemeister 3b

In[∘]:= $\text{lhs} = \int \mathbb{E}[\pi_i p_i + \pi_j p_j + \pi_k p_k + \mathcal{L}[1, i, j] + \mathcal{L}[1, i+1, k] + \mathcal{L}[1, j+1, k+1] + O[\epsilon]^2]$
$d\{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}\}$

Out[∘]=
$$\frac{1}{64\pi^6}\mathbb{E}\left[\left(T^2 p_{2+i}\pi_i - T p_{2+j}(-\pi_i + T\pi_i - \pi_j) + p_{2+k}(\pi_i - T\pi_i + \pi_j - T\pi_j + \pi_k)\right) - \frac{3\epsilon}{2} + O[\epsilon]^2\right]$$

In[∘]:= $\text{rhs} = \int \mathbb{E}[\pi_i p_i + \pi_j p_j + \pi_k p_k + \mathcal{L}[1, j, k] + \mathcal{L}[1, i, k+1] + \mathcal{L}[1, i+1, j+1] + O[\epsilon]^2]$
$d\{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}\}$

Out[∘]=
$$\frac{\mathbb{E}\left[\left(T^2 p_{2+i}\pi_i - T p_{2+j}(-\pi_i + T\pi_i - \pi_j) + p_{2+k}(\pi_i - T\pi_i + \pi_j - T\pi_j + \pi_k)\right) - \frac{3\epsilon}{2} + O[\epsilon]^2\right]}{64\pi^6}$$

In[ ]:=
```
lhs == rhs
```
Out[ ]=
```
True
```